

NAG Library Routine Document

S14CCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

S14CCF computes values for the incomplete beta function $I_x(a, b)$ and its complement $1 - I_x(a, b)$.

2 Specification

```
SUBROUTINE S14CCF (A, B, X, W, W1, IFAIL)
  INTEGER          IFAIL
  REAL (KIND=nag_wp) A, B, X, W, W1
```

3 Description

S14CCF evaluates the incomplete beta function and its complement in the normalized form

$$I_x(a, b) = \frac{1}{B(a, b)} \int_0^x t^{a-1} (1-t)^{b-1} dt$$

$$1 - I_x(a, b) = I_y(b, a), \text{ where } y = 1 - x,$$

with

$$0 \leq x \leq 1,$$

$$a \geq 0 \text{ and } b \geq 0,$$

and the beta function $B(a, b)$ is defined as $B(a, b) = \int_0^1 t^{a-1} (1-t)^{b-1} dt = \frac{\Gamma(a)\Gamma(b)}{\Gamma(a+b)}$ where $\Gamma(y)$ is the gamma function.

Several methods are used to evaluate the functions depending on the arguments a , b and x . The methods include Wise's asymptotic expansion (see Wise (1950)) when $a > b$, continued fraction derived by DiDonato and Morris (1992) when $a, b > 1$, and power series when $b \leq 1$ or $b \times x \leq 0.7$. When both a and b are large, specifically $a, b \geq 15$, the DiDonato and Morris (1992) asymptotic expansion is employed for greater efficiency.

Once either $I_x(a, b)$ or $I_y(b, a)$ is computed, the other is obtained by subtraction from 1. In order to avoid loss of relative precision in this subtraction, the smaller of $I_x(a, b)$ and $I_y(b, a)$ is computed first.

S14CCF is derived from BRATIO in DiDonato and Morris (1992).

4 References

DiDonato A R and Morris A H (1992) Algorithm 708: Significant digit computation of the incomplete beta function ratios *ACM Trans. Math. Software* **18** 360–373

Wise M E (1950) The incomplete beta function as a contour integral and a quickly converging series for its inverse *Biometrika* **37** 208–218

5 Parameters

- 1: A – REAL (KIND=nag_wp) Input
On entry: the argument a of the function.
Constraint: $A \geq 0.0$.
- 2: B – REAL (KIND=nag_wp) Input
On entry: the argument b of the function.
Constraints:
 $B \geq 0.0$;
 either $B \neq 0.0$ or $A \neq 0.0$.
- 3: X – REAL (KIND=nag_wp) Input
On entry: x , upper limit of integration.
Constraints:
 $0.0 \leq X \leq 1.0$;
 either $X \neq 0.0$ or $A \neq 0.0$;
 either $1 - X \neq 0.0$ or $B \neq 0.0$.
- 4: W – REAL (KIND=nag_wp) Output
On exit: the value of the incomplete beta function $I_x(a, b)$ evaluated from zero to x .
- 5: W1 – REAL (KIND=nag_wp) Output
On exit: the value of the complement of the incomplete beta function $1 - I_x(a, b)$, i.e., the incomplete beta function evaluated from x to one.
- 6: IFAIL – INTEGER Input/Output
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, A = $\langle value \rangle$.
 Constraint: $A \geq 0.0$.

On entry, B = $\langle value \rangle$.
 Constraint: $B \geq 0.0$.

IFAIL = 2

On entry, A and B were zero.
Constraint: A or B must be nonzero.

IFAIL = 3

On entry, $X = \langle \text{value} \rangle$.
Constraint: $0.0 \leq X \leq 1.0$.

IFAIL = 4

On entry, X and A were zero.
Constraint: X or A must be nonzero.

IFAIL = 5

On entry, $1.0 - X$ and B were zero.
Constraint: $1.0 - X$ or B must be nonzero.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.
See Section 3.8 in the Essential Introduction for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.
See Section 3.7 in the Essential Introduction for further information.

IFAIL = -999

Dynamic memory allocation failed.
See Section 3.6 in the Essential Introduction for further information.

7 Accuracy

S14CCF is designed to maintain relative accuracy for all arguments. For very tiny results (of the order of *machine precision* or less) some relative accuracy may be lost – loss of three or four decimal places has been observed in experiments. For other arguments full relative accuracy may be expected.

8 Parallelism and Performance

Not applicable.

9 Further Comments

None.

10 Example

This example reads values of the arguments a and b from a file, evaluates the function and its complement for 10 different values of x and prints the results.

10.1 Program Text

```

Program s14ccfe

!      S14CCF Example Program Text
!      Mark 25 Release. NAG Copyright 2014.

!      .. Use Statements ..
Use nag_library, Only: nag_wp, s14ccf
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: a, b, w, w1, x
Integer                    :: i, ifail
!      .. Executable Statements ..
Write (nout,*) 'S14CCF Example Program Results'
Write (nout,*)
Write (nout,*) &
'  A      B      X          Ix(A,B)          1-Ix(A,B)'
Write (nout,*)

!      Skip heading in data file
Read (nin,*)

Do
  Read (nin,*,Iostat=i) a, b, x
  If (i/=0) Exit

  ifail = -1
  Call s14ccf(a,b,x,w,w1,ifail)

  If (ifail==0) Then
    Write (nout,99999) a, b, x, w, w1
  End If
End Do

99999 Format (3F6.2,1P,E17.4,7X,E17.4)
End Program s14ccfe

```

10.2 Program Data

```

S14CCF Example Program Data
5.3  10.1  0.01
5.3  10.1  0.02
5.3  10.1  0.03
5.3  10.1  0.04
5.3  10.1  0.05
5.3  10.1  0.06
5.3  10.1  0.07
5.3  10.1  0.08
5.3  10.1  0.09
5.3  10.1  0.10 : A,B,X

```

10.3 Program Results

S14CCF Example Program Results

A	B	X	Ix(A,B)	1-Ix(A,B)
5.30	10.10	0.01	6.4755E-08	1.0000E+00
5.30	10.10	0.02	2.3613E-06	1.0000E+00
5.30	10.10	0.03	1.8734E-05	9.9998E-01
5.30	10.10	0.04	7.9575E-05	9.9992E-01
5.30	10.10	0.05	2.3997E-04	9.9976E-01

5.30	10.10	0.06	5.8255E-04	9.9942E-01
5.30	10.10	0.07	1.2174E-03	9.9878E-01
5.30	10.10	0.08	2.2797E-03	9.9772E-01
5.30	10.10	0.09	3.9249E-03	9.9608E-01
5.30	10.10	0.10	6.3236E-03	9.9368E-01
